

3. Tracé de courbes avec Scilab

Sommaire

3.1. Nuage de points

- Evolution des termes d'une suite

3.2. Courbe d'une fonction $y=f(x)$

3.3. Fonction définie par morceaux

3.4. Courbe en 3 dimensions d'une fonction $z=f(x, y)$

3.1. Nuage de points

Voici un exemple de relevé de températures au cours d'une journée :

t (heure)	0	6	10	12	14	16	18	20	24
T (°C)	12,4	10,8	18,1	20,7	22,8	22,0	19,5	16,8	13,2

Créons deux vecteurs :

```
--> t=[0 6 10 12 14 16 18 20 24]
t =
  0.    6.   10.   12.   14.   16.   18.   20.   24.

--> T=[12.4 10.8 18.1 20.7 22.8 22 19.5 16.8 13.2]
T =
  12.4   10.8   18.1   20.7   22.8   22.   19.5   16.8   13.2


--> plot(t,T)

--> xlabel("heure")

--> ylabel("Température (°C)")

--> title("Relevé de température")
```

Une fenêtre s'ouvre avec la figure (Figure n°0) :



- **Options**

- ✓ Type de traits

```
"-"      (trait plein par défaut)
"-."    pointillé
"--"    pointillé
":"     pointillé
```

- ✓ Type de points

```
(pas de point par défaut)
"."
"+"
"o"
"x"
"*"
```


- ✓ Couleur

```
"b" = blue (par défaut)
"k" = black
"r" = red
"g" = green
"c" = cyan
"m" = magenta
```

```
"y" = yellow
```

```
// pour effacer la figure  
--> clf()  
--> plot(t,T,"--or")
```

Résultat (image à insérer) :



```
// pour créer une nouvelle figure (Figure n°1)  
--> clf()  
--> plot(t,T,"+m")
```

Résultat (image à insérer) :



- **Evolution des termes d'une suite**

On s'intéresse à une suite géométrique :

$u_{n+1} = 0,8 u_n$
avec $u_0 = 5$


On veut tracer la courbe d'évolution de cette suite.
Pour cela, il faut calculer les termes de la suite et les stocker dans un vecteur.

```
// programme suite01.sce  
// création du vecteur u et assignation du premier terme  
// attention : le 1er indice est 1  
// u(1) représente donc u0  
u(1)=5  
// calcul des N premiers termes  
N=20  
for i=1:(N-1)  
    u(i+1)=0.8*u(i)  
end  
// affichage des N premiers termes  
disp(u)  
  
// figure  
clf()  
// axe des abscisses décalé pour commencer à 0  
n=0:(N-1)  
plot(n,u,"or")  
xlabel("n")  
ylabel("un")  
title("Suite géométrique")
```

Résultat :

-->

5.
4.
3.2
2.56
2.048
1.6384
1.31072
1.048576
0.8388608
0.6710886
0.5368709
0.4294967
0.3435974
0.2748779
0.2199023
0.1759219
0.1407375
0.1125900
0.0900720
0.0720576




Exercice 03-01

Tracer la courbe d'évolution de cette suite :

$$u_{n+1} = \frac{u_n}{2} + \frac{1}{u_n}$$

avec $u_0 = \frac{1}{2}$

Résultat (image à insérer) :



3.2. Courbe d'une fonction $y=f(x)$


On désire représenter la courbe de la fonction $f(x) = (2x^2 + x)e^{-x}$ dans l'intervalle $[-1, 10]$

```
// programme courbe01.sce
function y=f(x)
    y=(2*x^2+x)*exp(-x)
endfunction

// axe des abscisses
xmin=-1
xmax=10
n=1000 // nombre de points
x=linspace(xmin,xmax,n)

clf()
plot(x, f)
// fplot2d(x, f)
xlabel("x")
ylabel("y")
title("Fonction f(x)=(2*x^2+x)*exp(-x)")
```

Résultat :



Remarque : la fonction `linspace(x1, x2, n)` retourne un vecteur ligne de n valeurs régulièrement réparties entre x1 et x2

```
--> linspace(0, 0.9, 10)
ans =
0.0    0.1    0.2    0.3    0.4    0.5    0.6    0.7    0.8    0.9
```

Cela donne le même résultat que :

```
--> 0:0.1:0.9
ans =
0.0    0.1    0.2    0.3    0.4    0.5    0.6    0.7    0.8    0.9
```

Exercice 03-02

On s'intéresse à une tension électrique dont l'évolution au cours du temps est donnée par la relation :

$$u(t) = 1 + \cos(100\pi t) - \frac{\sin(300\pi t)}{3}$$

u en volts

t en secondes

Tracer la courbe $u(t)$ sur l'intervalle $[0, 0,06 \text{ s}]$

Résultat (image à insérer)

.....

3.3. Fonction définie par morceaux

- Exemple : champ magnétique créé par un câble électrique

$$B(r) = \frac{\mu_0 I}{2\pi R^2} r \quad \text{pour } 0 \leq r < R$$

$$B(r) = \frac{\mu_0 I}{2\pi r} \quad \text{pour } r \geq R$$

B est l'intensité du champ magnétique (en tesla)

r désigne la distance à l'axe du câble ($r \geq 0$)

R est le rayon du câble

I est l'intensité du courant électrique (en ampère)

$\mu_0 = 4\pi 10^{-7} \text{ H/m}$ est une constante universelle (perméabilité magnétique du vide)


```
// programme cable01.sce

// variables globales
R = 1.5 // rayon du câble en mm
I = 10.0 // intensité du courant en ampère
mu0 = 4*pi*1e-7 // perméabilité magnétique du vide

// fonction définie par morceaux
function B=B(r)
    // r distance en mm (positive)
    if r >= R then
        // à l'extérieur du câble
        B=mu0*I/(2*pi*r*1e-3)
    elseif r >= 0
        // à l'intérieur
        B=mu0*I*r/(2*pi*R*R*1e-3)
    end
endfunction

// figure
clf()
d=linspace(0,20,1000)
// plot(d,B) // scilab version 5
fplot2d(d,B) // scilab version 6
xlabel("Distance (mm)")
ylabel("Intensité du champ magnétique (tesla)")
title("Champ magnétique créé par un câble électrique")
```

Résultat :



Tracer la courbe $B(r)$ pour un câble haute tension de rayon 20 mm parcouru par un courant de 3000 A, sur l'intervalle $[0, 500 \text{ mm}]$:

.....

Exercice 03-03 Fonction échelon

On considère la fonction h définie par :

$$h(t) = 0 \text{ quand } t < 0$$

$$h(t) = 1 \text{ quand } t \geq 0$$

avec t le temps

Tracé la courbe $h(t)$ dans l'intervalle $[-1, 5 \text{ s}]$

.....

3.4. Courbe en 3 dimensions d'une fonction $z=f(x,y)$

On considère la fonction f définie par :


$$z=f(x,y)=y \cdot \exp(-x^2-y^2)$$

Nous allons faire une représentation graphique de cette fonction dans un espace à 3 dimensions.

```
// programme courbe3d_01.sce
function z=f(x, y)
    z=y*exp(-x^2-y^2)
endfunction

// figure
x=linspace(-2,2,100) // axe x 100 points
y=linspace(-2,2,100) // axe y 100 points
z=feval(x,y,f)'      // grille 100x100 = 10000 points
clf()
surf(x,y,z)
xlabel("x")
ylabel("y")
zlabel("z=f(x,y)")
title("Fonction z=y*exp(-x^2-y^2)")
```

Résultat :




Remarque : clic droit avec la souris pour un effet de rotation en 3d

Pour afficher les courbes de niveau (en 2d), ajouter ces lignes au programme précédent :

```
// courbes de niveau
scf()
contour(x,y,z',10)
xlabel("x")
ylabel("y")
title("Courbes de niveau")
```

Résultat :



- Licence

Cette œuvre est mise à disposition selon les termes de la **Licence Creative Commons** :

- Attribution
- Pas d'utilisation commerciale
- Partage dans les mêmes conditions
- version 3.0 France

<http://creativecommons.org/licenses/by-nc-sa/3.0/fr/>

© **Fabrice Sincère**

- Page d'accueil : <http://fabrice.sincere.pagesperso-orange.fr/>
<http://fabrice.sincere.free.fr/>
- Mail : fabrice.sincere@wanadoo.fr
fabrice.sincere@ac-grenoble.fr